
NAG Fortran Library Routine Document

D03PEF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

D03PEF integrates a system of linear or nonlinear, first-order, time-dependent partial differential equations
(PDEs) in one space variable. The spatial discretization is performed using the Keller box scheme and the
method of lines is employed to reduce the PDEs to a system of ordinary differential equations (ODEs).
The resulting system is solved using a Backward Differentiation Formula (BDF) method.

2 Specification

SUBROUTINE D03PEF (NPDE, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, NLEFT,
1 ACC, RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE,
2 IND, IFAIL)

INTEGER NPDE, NPTS, NLEFT, LRSAVE, ISAVE(LISAVE), LISAVE,
1 ITASK, ITRACE, IND, IFAIL
double precision TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC, RSAVE(LRSAVE)
EXTERNAL PDEDEF, BNDARY

3 Description

D03PEF integrates the system of first-order PDEs

Gi x; t;U ;Ux;Utð Þ ¼ 0, i ¼ 1; 2; . . . ;NPDE. ð1Þ
In particular the functions Gi must have the general form

Gi ¼
XNPDE

j¼1

Pi;j

@Uj

@t
þ Qi, i ¼ 1; 2; . . . ;NPDE, a � x � b; t � t0, ð2Þ

where Pi;j and Qi depend on x, t, U , Ux and the vector U is the set of solution values

U x; tð Þ ¼ U 1 x; tð Þ; . . . ;UNPDE x; tð Þ
h iT

, ð3Þ

and the vector Ux is its partial derivative with respect to x. Note that Pi;j and Qi must not depend on
@U

@t
.

The integration in time is from t0 to tout, over the space interval a � x � b, where a ¼ x1 and b ¼ xNPTS
are the leftmost and rightmost points of a user-defined mesh x1; x2; . . . ; xNPTS. The mesh should be chosen

in accordance with the expected behaviour of the solution.

The PDE system which is defined by the functions Gi must be specified in a (sub)program PDEDEF
supplied by you.

The initial values of the functions U x; tð Þ must be given at t ¼ t0. For a first-order system of PDEs, only
one boundary condition is required for each PDE component Ui. The NPDE boundary conditions are
separated into na at the left-hand boundary x ¼ a, and nb at the right-hand boundary x ¼ b, such that
na þ nb ¼ NPDE. The position of the boundary condition for each component should be chosen with care;
the general rule is that if the characteristic direction of Ui at the left-hand boundary (say) points into the
interior of the solution domain, then the boundary condition for Ui should be specified at the left-hand
boundary. Incorrect positioning of boundary conditions generally results in initialization or integration
difficulties in the underlying time integration routines.

The boundary conditions have the form:

GL
i x; t;U ;Utð Þ ¼ 0 at x ¼ a, i ¼ 1; 2; . . . ; na ð4Þ

at the left-hand boundary, and

GR
i x; t;U ;Utð Þ ¼ 0 at x ¼ b, i ¼ 1; 2; . . . ; nb ð5Þ

at the right-hand boundary.

Note that the functions GL
i and GR

i must not depend on Ux, since spatial derivatives are not determined
explicitly in the Keller box scheme (see Keller (1970)). If the problem involves derivative (Neumann)
boundary conditions then it is generally possible to restate such boundary conditions in terms of

permissible variables. Also note that GL
i and GR

i must be linear with respect to time derivatives, so that the
boundary conditions have the general form

XNPDE

j¼1

EL
i;j

@Uj

@t
þ SLi ¼ 0, i ¼ 1; 2; . . . ; na ð6Þ

at the left-hand boundary, and

XNPDE

j¼1

ER
i;j

@Uj

@t
þ SRi ¼ 0, i ¼ 1; 2; . . . ; nb ð7Þ

at the right-hand boundary, where EL
i;j, E

R
i;j, S

L
i , and SRi depend on x, t and U only.

The boundary conditions must be specified in a (sub)program BNDARY provided by you.

The problem is subject to the following restrictions:

(i) t0 < tout, so that integration is in the forward direction;

(ii) Pi;j and Qi must not depend on any time derivatives;

(iii) The evaluation of the function Gi is done at the mid-points of the mesh intervals by calling the
(sub)program PDEDEF for each mid-point in turn. Any discontinuities in the function must therefore
be at one or more of the mesh points x1; x2; . . . ; xNPTS;

(iv) At least one of the functions Pi;j must be non-zero so that there is a time derivative present in the
problem.

In this method of lines approach the Keller box scheme (see Keller (1970)) is applied to each PDE in the
space variable only, resulting in a system of ODEs in time for the values of Ui at each mesh point. In total
there are NPDE� NPTS ODEs in the time direction. This system is then integrated forwards in time
using a BDF method.

4 References

Berzins M (1990) Developments in the NAG Library software for parabolic equations Scientific Software
Systems (ed J C Mason and M G Cox) 59–72 Chapman and Hall

Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using
the method of lines and differential-algebraic integrators Appl. Numer. Math. 5 375–397

Keller H B (1970) A new difference scheme for parabolic problems Numerical Solutions of Partial
Differential Equations (ed J Bramble) 2 327–350 Academic Press

Pennington S V and Berzins M (1994) New NAG Library software for first-order partial differential
equations ACM Trans. Math. Softw. 20 63–99

5 Parameters

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system to be solved.

Constraint: NPDE � 1.

2: TS – double precision Input/Output

On entry: the initial value of the independent variable t.

Constraint: TS < TOUT.

On exit: the value of t corresponding to the solution values in U. Normally TS ¼ TOUT.

3: TOUT – double precision Input

On entry: the final value of t to which the integration is to be carried out.

4: PDEDEF – SUBROUTINE, supplied by the user. External Procedure

PDEDEF must compute the functions Gi which define the system of PDEs. PDEDEF is called
approximately midway between each pair of mesh points in turn by D03PEF.

Its specification is:

SUBROUTINE PDEDEF (NPDE, T, X, U, UT, UX, RES, IRES)

INTEGER NPDE, IRES
double precision T, X, U(NPDE), UT(NPDE), UX(NPDE), RES(NPDE)

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – double precision Input

On entry: the current value of the independent variable t.

3: X – double precision Input

On entry: the current value of the space variable x.

4: UðNPDEÞ – double precision array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ, for i ¼ 1; 2; . . . ;NPDE.

5: UTðNPDEÞ – double precision array Input

On entry: UTðiÞ contains the value of the component
@Ui x; tð Þ

@t
, for i ¼ 1; 2; . . . ;NPDE.

6: UXðNPDEÞ – double precision array Input

On entry: UXðiÞ contains the value of the component
@Ui x; tð Þ

@x
, for i ¼ 1; 2; . . . ;NPDE.

7: RESðNPDEÞ – double precision array Output

On exit: RESðiÞ must contain the ith component of G, for i ¼ 1; 2; . . . ;NPDE, where G is
defined as

Gi ¼
XNPDE

j¼1

Pi;j

@Uj

@t
, ð8Þ

i.e., only terms depending explicitly on time derivatives, or

Gi ¼
XNPDE

j¼1

Pi;j

@Uj

@t
þ Qi, ð9Þ

i.e., all terms in equation (2).

The definition of G is determined by the input value of IRES.

8: IRES – INTEGER Input/Output

On entry: the form of Gi that must be returned in the array RES. If IRES ¼ �1, then
equation (8) must be used. If IRES ¼ 1, then equation (9) must be used.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions, as described below:

IRES ¼ 2

Indicates to the integrator that control should be passed back immediately to the
calling (sub)program with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3

Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PEF returns to the calling (sub)program with the error indicator
set to IFAIL ¼ 4.

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PEF is called.
Parameters denoted as Input must not be changed by this procedure.

5: BNDARY – SUBROUTINE, supplied by the user. External Procedure

BNDARY must compute the functions GL
i and GR

i which define the boundary conditions as in
equations (4) and (5).

Its specification is:

SUBROUTINE BNDARY (NPDE, T, IBND, NOBC, U, UT, RES, IRES)

INTEGER NPDE, IBND, NOBC, IRES
double precision T, U(NPDE), UT(NPDE), RES(NOBC)

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – double precision Input

On entry: the current value of the independent variable t.

3: IBND – INTEGER Input

On entry: determines the position of the boundary conditions.

IBND ¼ 0

BNDARY must compute the left-hand boundary condition at x ¼ a.

IBND 6¼ 0

Indicates that BNDARY must compute the right-hand boundary condition at x ¼ b.

4: NOBC – INTEGER Input

On entry: specifies the number of boundary conditions at the boundary specified by IBND.

5: UðNPDEÞ – double precision array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ at the boundary specified by
IBND, for i ¼ 1; 2; . . . ;NPDE.

6: UTðNPDEÞ – double precision array Input

On entry: UTðiÞ contains the value of the component
@Ui x; tð Þ

@t
at the boundary specified

by IBND, for i ¼ 1; 2; . . . ;NPDE.

7: RESðNOBCÞ – double precision array Output

On exit: RESðiÞ must contain the ith component of GL or GR, depending on the value of

IBND, for i ¼ 1; 2; . . . ;NOBC, where GL is defined as

GL
i ¼

XNPDE

j¼1

EL
i;j

@Uj

@t
, ð10Þ

i.e., only terms depending explicitly on time derivatives, or

GL
i ¼

XNPDE

j¼1

EL
i;j

@Uj

@t
þ SLi , ð11Þ

i.e., all terms in equation (6), and similarly for GR
i .

The definitions of GL and GR are determined by the input value of IRES.

8: IRES – INTEGER Input/Output

On entry: the form GL
i (or GR

i ) that must be returned in the array RES. If IRES ¼ �1,
then equation (10) must be used. If IRES ¼ 1, then equation (11) must be used.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions, as described below:

IRES ¼ 2

Indicates to the integrator that control should be passed back immediately to the
calling (sub)program with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3

Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PEF returns to the calling (sub)program with the error indicator
set to IFAIL ¼ 4.

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PEF is called.
Parameters denoted as Input must not be changed by this procedure.

6: UðNPDE,NPTSÞ – double precision array Input/Output

On entry: the initial values of U x; tð Þ at t ¼ TS and the mesh points XðjÞ, for j ¼ 1; 2; . . . ;NPTS.

On exit: Uði; jÞ will contain the computed solution at t ¼ TS.

7: NPTS – INTEGER Input

On entry: the number of mesh points in the interval a; b½ �.
Constraint: NPTS � 3.

8: XðNPTSÞ – double precision array Input

On entry: the mesh points in the spatial direction. Xð1Þ must specify the left-hand boundary, a, and
XðNPTSÞ must specify the right-hand boundary, b.

Constraint: Xð1Þ < Xð2Þ < � � � < XðNPTSÞ.

9: NLEFT – INTEGER Input

On entry: the number na of boundary conditions at the left-hand mesh point Xð1Þ.
Constraint: 0 � NLEFT � NPDE.

10: ACC – double precision Input

On entry: a positive quantity for controlling the local error estimate in the time integration. If E i; jð Þ
is the estimated error for Ui at the jth mesh point, the error test is:

E i; jð Þj j ¼ ACC� 1:0þ Uði; jÞj jð Þ.
Constraint: ACC > 0:0.

11: RSAVEðLRSAVEÞ – double precision array Communication Array

If IND ¼ 0, RSAVE need not be set on entry.

If IND ¼ 1, RSAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration.

12: LRSAVE – INTEGER Input

On entry: the dimension of the array RSAVE as declared in the (sub)program from which D03PEF
is called.

Constraint:
LRSAVE � 4� NPDEþ NLEFTþ 14ð Þ � NPDE� NPTSþ 3� NPDEþ 21ð Þ � NPDEþ
7� NPTSþ 54.

13: ISAVEðLISAVEÞ – INTEGER array Communication Array

If IND ¼ 0, ISAVE need not be set on entry.

If IND ¼ 1, ISAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration. In particular:

ISAVEð1Þ
Contains the number of steps taken in time.

ISAVEð2Þ
Contains the number of residual evaluations of the resulting ODE system used. One such
evaluation involves computing the PDE functions at all the mesh points, as well as one
evaluation of the functions in the boundary conditions.

ISAVEð3Þ
Contains the number of Jacobian evaluations performed by the time integrator.

ISAVEð4Þ
Contains the order of the last backward differentiation formula method used.

ISAVEð5Þ
Contains the number of Newton iterations performed by the time integrator. Each iteration
involves an ODE residual evaluation followed by a back-substitution using the LU
decomposition of the Jacobian matrix.

14: LISAVE – INTEGER Input

On entry: the dimension of the array ISAVE as declared in the (sub)program from which D03PEF is
called.

Constraint: LISAVE � NPDE� NPTSþ 24.

15: ITASK – INTEGER Input

On entry: specifies the task to be performed by the ODE integrator.

ITASK ¼ 1

Normal computation of output values U at t ¼ TOUT.

ITASK ¼ 2

Take one step and return.

ITASK ¼ 3

Stop at the first internal integration point at or beyond t ¼ TOUT.

Constraint: 1 � ITASK � 3.

16: ITRACE – INTEGER Input

On entry: the level of trace information required from D03PEF and the underlying ODE solver as
follows:

ITRACE � �1

No output is generated.

ITRACE ¼ 0

Only warning messages from the PDE solver are printed on the current error message unit
(see X04AAF).

ITRACE ¼ 1

Output from the underlying ODE solver is printed on the current advisory message unit (see
X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the
time integration during the computation of the ODE system.

ITRACE ¼ 2

Output from the underlying ODE solver is similar to that produced when ITRACE ¼ 1,
except that the advisory messages are given in greater detail.

ITRACE � 3

Output from the underlying ODE solver is similar to that produced when ITRACE ¼ 2,
except that the advisory messages are given in greater detail.

You are advised to set ITRACE ¼ 0, unless you are experienced with sub-chapter D02M/N.

17: IND – INTEGER Input/Output

On entry: must be set to 0 or 1.

IND ¼ 0

Starts or restarts the integration in time.

IND ¼ 1

Continues the integration after an earlier exit from the routine. In this case, only the
parameters TOUT and IFAIL should be reset between calls to D03PEF.

Constraint: 0 � IND � 1.

On exit: IND ¼ 1.

18: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Chapter P01 for details.

On exit: IFAIL ¼ 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, TOUT � TS,
or TOUT� TSð Þ is too small,
or ITASK 6¼ 1, 2 or 3,
or mesh points XðiÞ are not ordered correctly,
or NPTS < 3,
or NPDE < 1,
or NLEFT is not in the range 0 to NPDE,
or ACC � 0:0,
or IND 6¼ 0 or 1,
or LRSAVE is too small,
or LISAVE is too small,
or D03PEF called initially with IND ¼ 1.

IFAIL ¼ 2

The underlying ODE solver cannot make any further progress, across the integration range from the
current point t ¼ TS with the supplied value of ACC. The components of U contain the computed
values at the current point t ¼ TS.

IFAIL ¼ 3

In the underlying ODE solver, there were repeated errors or corrector convergence test failures on an
attempted step, before completing the requested task. The problem may have a singularity or ACC
is too small for the integration to continue. Incorrect positioning of boundary conditions may also
result in this error. Integration was successful as far as t ¼ TS.

IFAIL ¼ 4

In setting up the ODE system, the internal initialization routine was unable to initialize the
derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in
the user-supplied (sub)program PDEDEF or BNDARY, when the residual in the underlying ODE
solver was being evaluated. Incorrect positioning of boundary conditions may also result in this
error.

IFAIL ¼ 5

In solving the ODE system, a singular Jacobian has been encountered. You should check their
problem formulation.

IFAIL ¼ 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in one of the user-
supplied (sub)programs PDEDEF or BNDARY. Integration was successful as far as t ¼ TS.

IFAIL ¼ 7

The value of ACC is so small that the routine is unable to start the integration in time.

IFAIL ¼ 8

In one of the user-supplied (sub)programs, PDEDEF or BNDARY, IRES was set to an invalid value.

IFAIL ¼ 9 (D02NNF)

A serious error has occurred in an internal call to the specified routine. Check problem specification
and all parameters and array dimensions. Setting ITRACE ¼ 1 may provide more information. If
the problem persists, contact NAG.

IFAIL ¼ 10

The required task has been completed, but it is estimated that a small change in ACC is unlikely to
produce any change in the computed solution. (Only applies when you are not operating in one step
mode, that is when ITASK 6¼ 2.)

IFAIL ¼ 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error description
may be directed to the current advisory message unit).

7 Accuracy

D03PEF controls the accuracy of the integration in the time direction but not the accuracy of the
approximation in space. The spatial accuracy depends on both the number of mesh points and on their
distribution in space. In the time integration only the local error over a single step is controlled and so the
accuracy over a number of steps cannot be guaranteed. You should therefore test the effect of varying the
accuracy parameter, ACC.

8 Further Comments

The Keller box scheme can be used to solve higher-order problems which have been reduced to first-order
by the introduction of new variables (see the example problem in D03PKF). In general, a second-order
problem can be solved with slightly greater accuracy using the Keller box scheme instead of a finite-
difference scheme (D03PCF=D03PCA or D03PHF=D03PHA for example), but at the expense of increased
CPU time due to the larger number of function evaluations required.

It should be noted that the Keller box scheme, in common with other central-difference schemes, may be
unsuitable for some hyperbolic first-order problems such as the apparently simple linear advection equation
Ut þ aUx ¼ 0, where a is a constant, resulting in spurious oscillations due to the lack of dissipation. This
type of problem requires a discretization scheme with upwind weighting (D03PFF for example), or the
addition of a second-order artificial dissipation term.

The time taken depends on the complexity of the system and on the accuracy requested.

D03 – Partial Differential Equations D03PEF

[NP3657/21] D03PEF.9



9 Example

This example is the simple first-order system

@U 1

@t
þ @U 1

@x
þ @U2

@x
¼ 0,

@U 2

@t
þ 4

@U 1

@x
þ @U2

@x
¼ 0,

for t 2 0; 1½ � and x 2 0; 1½ �.
The initial conditions are

U 1 x; 0ð Þ ¼ exp xð Þ, U 2 x; 0ð Þ ¼ sin xð Þ,
and the Dirichlet boundary conditions for U 1 at x ¼ 0 and U 2 at x ¼ 1 are given by the exact solution:

U 1 x; tð Þ ¼ 1
2 exp xþ tð Þ þ exp x� 3tð Þf g þ 1

4 sin x� 3tð Þ � sin xþ tð Þf g,

U 2 x; tð Þ ¼ exp x� 3tð Þ � exp xþ tð Þ þ 1
2 sin xþ tð Þ þ sin x� 3tð Þf g.

9.1 Program Text

* D03PEF Example Program Text
* Mark 16 Release. NAG Copyright 1993.
* .. Parameters ..

INTEGER NOUT
PARAMETER (NOUT=6)
INTEGER NPDE, NPTS, NLEFT, NEQN, NIW, NWKRES, NW
PARAMETER (NPDE=2,NPTS=41,NLEFT=1,NEQN=NPDE*NPTS,

+ NIW=NEQN+24,NWKRES=NPDE*(NPTS+21+3*NPDE)
+ +7*NPTS+4,NW=11*NEQN+(4*NPDE+NLEFT+2)
+ *NEQN+50+NWKRES)

* .. Local Scalars ..
DOUBLE PRECISION ACC, TOUT, TS
INTEGER I, IFAIL, IND, IT, ITASK, ITRACE

* .. Local Arrays ..
DOUBLE PRECISION EU(NPDE,NPTS), U(NPDE,NPTS), W(NW), X(NPTS)
INTEGER IW(NIW)

* .. External Subroutines ..
EXTERNAL BNDARY, D03PEF, EXACT, PDEDEF, UINIT

* .. Executable Statements ..
WRITE (NOUT,*) ’D03PEF Example Program Results’
ITRACE = 0
ACC = 0.1D-5
WRITE (NOUT,99997) ACC, NPTS

*
* Set spatial-mesh points
*

DO 20 I = 1, NPTS
X(I) = (I-1.0D0)/(NPTS-1.0D0)

20 CONTINUE
WRITE (NOUT,99999) X(5), X(13), X(21), X(29), X(37)

*
IND = 0
ITASK = 1

*
CALL UINIT(NPDE,NPTS,X,U)

*
* Loop over output value of t

TS = 0.0D0
TOUT = 0.0D0
DO 40 IT = 1, 5

TOUT = 0.2D0*IT
IFAIL = -1

*
CALL D03PEF(NPDE,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,NLEFT,ACC,W,NW,

+ IW,NIW,ITASK,ITRACE,IND,IFAIL)
*

* Check against the exact solution
*

CALL EXACT(TOUT,NPDE,NPTS,X,EU)
*

WRITE (NOUT,99998) TS
WRITE (NOUT,99995) U(1,5), U(1,13), U(1,21), U(1,29), U(1,37)
WRITE (NOUT,99994) EU(1,5), EU(1,13), EU(1,21), EU(1,29),

+ EU(1,37)
WRITE (NOUT,99993) U(2,5), U(2,13), U(2,21), U(2,29), U(2,37)
WRITE (NOUT,99992) EU(2,5), EU(2,13), EU(2,21), EU(2,29),

+ EU(2,37)
40 CONTINUE

WRITE (NOUT,99996) IW(1), IW(2), IW(3), IW(5)
STOP

*
99999 FORMAT (’ X ’,5F10.4,/)
99998 FORMAT (’ T = ’,F5.2)
99997 FORMAT (//’ Accuracy requirement =’,E10.3,’ Number of points = ’,

+ I3,/)
99996 FORMAT (’ Number of integration steps in time = ’,I6,/’ Number o’,

+ ’f function evaluations = ’,I6,/’ Number of Jacobian eval’,
+ ’uations =’,I6,/’ Number of iterations = ’,I6)

99995 FORMAT (’ Approx U1’,5F10.4)
99994 FORMAT (’ Exact U1’,5F10.4)
99993 FORMAT (’ Approx U2’,5F10.4)
99992 FORMAT (’ Exact U2’,5F10.4,/)

END
*

SUBROUTINE UINIT(NPDE,NPTS,X,U)
* Routine for PDE initial values
* .. Scalar Arguments ..

INTEGER NPDE, NPTS
* .. Array Arguments ..

DOUBLE PRECISION U(NPDE,NPTS), X(NPTS)
* .. Local Scalars ..

INTEGER I
* .. Intrinsic Functions ..

INTRINSIC EXP, SIN
* .. Executable Statements ..

DO 20 I = 1, NPTS
U(1,I) = EXP(X(I))
U(2,I) = SIN(X(I))

20 CONTINUE
RETURN
END

*
SUBROUTINE PDEDEF(NPDE,T,X,U,UDOT,DUDX,RES,IRES)

* .. Scalar Arguments ..
DOUBLE PRECISION T, X
INTEGER IRES, NPDE

* .. Array Arguments ..
DOUBLE PRECISION DUDX(NPDE), RES(NPDE), U(NPDE), UDOT(NPDE)

* .. Executable Statements ..
IF (IRES.EQ.-1) THEN

RES(1) = UDOT(1)
RES(2) = UDOT(2)

ELSE
RES(1) = UDOT(1) + DUDX(1) + DUDX(2)
RES(2) = UDOT(2) + 4.0D0*DUDX(1) + DUDX(2)

END IF
RETURN
END

*
SUBROUTINE BNDARY(NPDE,T,IBND,NOBC,U,UDOT,RES,IRES)

* .. Scalar Arguments ..
DOUBLE PRECISION T
INTEGER IBND, IRES, NOBC, NPDE

* .. Array Arguments ..
DOUBLE PRECISION RES(NOBC), U(NPDE), UDOT(NPDE)

* .. Intrinsic Functions ..
INTRINSIC EXP, SIN

* .. Executable Statements ..
IF (IBND.EQ.0) THEN

IF (IRES.EQ.-1) THEN
RES(1) = 0.0D0

ELSE
RES(1) = U(1) - 0.5D0*(EXP(T)+EXP(-3.0D0*T)) -

+ 0.25D0*(SIN(-3.0D0*T)-SIN(T))
END IF

ELSE
IF (IRES.EQ.-1) THEN

RES(1) = 0.0D0
ELSE

RES(1) = U(2) - EXP(1.0D0-3.0D0*T) + EXP(1.0D0+T) -
+ 0.5D0*(SIN(1.0D0-3.0D0*T)+SIN(1.0D0+T))

END IF
END IF
RETURN
END

*
SUBROUTINE EXACT(T,NPDE,NPTS,X,U)

* Exact solution (for comparison purposes)
* .. Scalar Arguments ..

DOUBLE PRECISION T
INTEGER NPDE, NPTS

* .. Array Arguments ..
DOUBLE PRECISION U(NPDE,NPTS), X(NPTS)

* .. Local Scalars ..
INTEGER I

* .. Intrinsic Functions ..
INTRINSIC EXP, SIN

* .. Executable Statements ..
DO 20 I = 1, NPTS

U(1,I) = 0.5D0*(EXP(X(I)+T)+EXP(X(I)-3.0D0*T)) +
+ 0.25D0*(SIN(X(I)-3.0D0*T)-SIN(X(I)+T))

U(2,I) = EXP(X(I)-3.0D0*T) - EXP(X(I)+T) + 0.5D0*(SIN(X(I)
+ -3.0D0*T)+SIN(X(I)+T))

20 CONTINUE
RETURN
END

9.2 Program Data

None.

9.3 Program Results

D03PEF Example Program Results

Accuracy requirement = 0.100E-05 Number of points = 41

X 0.1000 0.3000 0.5000 0.7000 0.9000

T = 0.20
Approx U1 0.7845 1.0010 1.2733 1.6115 2.0281
Exact U1 0.7845 1.0010 1.2733 1.6115 2.0281
Approx U2 -0.8352 -0.8159 -0.8367 -0.9128 -1.0609
Exact U2 -0.8353 -0.8160 -0.8367 -0.9129 -1.0609

T = 0.40
Approx U1 0.6481 0.8533 1.1212 1.4627 1.8903
Exact U1 0.6481 0.8533 1.1212 1.4627 1.8903
Approx U2 -1.5216 -1.6767 -1.8934 -2.1917 -2.5944
Exact U2 -1.5217 -1.6767 -1.8935 -2.1917 -2.5945

T = 0.60
Approx U1 0.6892 0.8961 1.1747 1.5374 1.9989
Exact U1 0.6892 0.8962 1.1747 1.5374 1.9989
Approx U2 -2.0047 -2.3434 -2.7677 -3.3002 -3.9680
Exact U2 -2.0048 -2.3436 -2.7678 -3.3003 -3.9680

T = 0.80
Approx U1 0.8977 1.1247 1.4320 1.8349 2.3514
Exact U1 0.8977 1.1247 1.4320 1.8349 2.3512
Approx U2 -2.3403 -2.8675 -3.5110 -4.2960 -5.2536
Exact U2 -2.3405 -2.8677 -3.5111 -4.2961 -5.2537

T = 1.00
Approx U1 1.2470 1.5206 1.8828 2.3528 2.9519
Exact U1 1.2470 1.5205 1.8829 2.3528 2.9518
Approx U2 -2.6229 -3.3338 -4.1998 -5.2505 -6.5218
Exact U2 -2.6232 -3.3340 -4.2001 -5.2507 -6.5219

Number of integration steps in time = 149
Number of function evaluations = 399
Number of Jacobian evaluations = 13
Number of iterations = 323

D03 – Partial Differential Equations D03PEF

